TI-80

Quadratic Formula Program

This program will display the solutions of a quadratic equation or the words "No Real Solution." To use the program, write the quadratic equation in general form and enter the values of *a, b,* and *c*.

```
PROGRAM:QUADRAT
:DISP "AX²+BX+C=0"
:INPUT "ENTER A",A
:INPUT "ENTER B",B
:INPUT "ENTER C",C
:B²−4AC→D
:IF D≥0
:THEN
:(-B+√(D)/(2A)→M
:DISP M
:(-B−√(D)/(2A)→N
:DISP N
:ELSE
:DISP "NO REAL SOLUTION"
:END
```

Graph Reflection Program

This program will graph a function *f* and its reflection in the line $y = x$. To use this program, enter the function in Y_1 and set a viewing rectangle.

```
PROGRAM:REFLECT
:47XMIN/63→YMIN
:47XMAX/63→YMAX
:XSCL→YSCL
:"X"→Y2
:DISPGRAPH
:(XMAX−XMIN)/62→I
:XMIN→X
:LBL A
:PT-ON(Y1,X)
:X+I→X
:If X>XMAX
:STOP
:GOTO A
```

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
PROGRAM:SOLVE
:DISP "AX+BY=C"
:INPUT "ENTER A",A
:INPUT "ENTER B",B
:INPUT "ENTER C",C
:DISP "DX+EY=F"
:INPUT "ENTER D",D
:INPUT "ENTER E",E
:INPUT "ENTER F",F
:IF AE−DB=0
:THEN
:DISP "NO UNIQUE"
:DISP "SOLUTION"
:ELSE
:(CE−BF)/(AE−DB)→X
:(AF−CD)/(AE−DB)→Y
:DISP X
:DISP Y
:END
```

Visualizing Row Operations Program not available

Evaluating an Algebraic Expression Program

This program can be used to evaluate an algebraic expression in one variable at several values of the variable. To use this program, enter an expression in Y1.

```
PROGRAM:EVALUAT
:LBL A
:INPUT "ENTER X",X
:DISP Y1
:GOTO A
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
PROGRAM:ADDVECT
:CLRDRAW
:DISP "ENTER(A,B)"
:INPUT "ENTER A",A
:INPUT "ENTER B",B
:DISP "ENTER (C,D)"
:INPUT "ENTER C",C
:INPUT "ENTER D",D
:LINE(0,0,A,B)
:LINE(0,0,C,D)
:A+C→E
:B+D→F
:LINE(0,0,E,F)
:LINE(A,B,E,F)
:LINE(C,D,E,F)
:PAUSE
```

Graphing a Sine Function Program

This program will simultaneously draw a unit circle and the corresponding points on the sine curve. After the circle and sine curve are drawn, you can connect the points on the unit circle with their corresponding points on the sine curve by pressing ENTER.

```
PROGRAM:SINESHO
:RADIAN
:CLRDRAW:FNOFF
:PARAM:SIMUL
:-2.25→XMIN
:π/2→XMAX
:3→XSCL
:-1.5→YMIN
:1.5→YMAX
:1→YSCL
:0→TMIN
:6.3→TMAX
:.15→TSTEP
:"-1.25+COS T"→X1T
:"SIN T"→Y1T
:"T/4"→X2T
:"SIN T"→Y2T
:DISPGRAPH
:FOR(N,1,12)
:Nπ/6.5→T
:"-1.25+COS T"→A
:SIN T→B
:T/4→C
:LINE(A,B,C,B)
:PAUSE
:END
:PAUSE:FUNC
:SEQUENTIAL:DISP
```

Finding the Angle Between Two Vectors Program

This program will sketch two vectors and calculate the measure of the angle between the vectors. Be sure to set an appropriate viewing rectangle.

```
:PROGRAM:VECANGL
:CLRHOME
:DEGREE
:DISP "ENTER (A,B)"
:INPUT "ENTER A",A
:INPUT "ENTER B",B
:CLRHOME
:DISP "ENTER (C,D)"
:INPUT "ENTER C",C
:INPUT "ENTER D",D
:LINE(0,0,A,B)
:LINE(0,0,C,D)
:PAUSE
:AC+BD→E
:√(A²+B²)→U
:√(C²+D²)→V
:COS⁻¹(E/(UV))→θ
:DISP "θ=", θ
:CLRDRAW
```

TI-81

Quadratic Formula Program

This program will display the solutions of a quadratic equation or the words "No Real Solution." To use the program, write the quadratic equation in general form and enter the values of *a, b,* and *c.*

```
Prgm1: QUADRAT
:Disp "ENTER A"
:Input A
:Disp "ENTER B"
:Input B
:Disp "ENTER C"
:Input C
:B^2−4AC→D
:If D<0
:Goto 1
:((-B+√(D)/(2A))→M
:Disp M
:((-B−√(D)/(2A))→N
:Disp N
:End
:Lbl 1
:Disp "NO REAL"
:Disp "SOLUTION"
:End
```

Graph Reflection Program

This program will graph a function *f* and its reflection in the line $y = x$. To use this program, enter the function in Y_1 and set a viewing rectangle.

```
Prgm2:REFLECT
:2Xmin/3→Ymin
:2Xmax/3→Ymax
:Xscl→Yscl
:"X"→Y2
:DispGraph
:(Xmax−Xmin)/95→I
:Xmin→X
:Lbl 1
:Pt-On(Y1,X)
:X+I→X
:If X>Xmax
:End
:Goto 1
```

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
Prgm3:SOLVE
:Disp "AX+BY=C"
:Input A
:Input B
:Input C
:Disp "DX+EY=F"
:Input D
:Input E
:Input F
:If AE−DB=0
:Goto 1
:(CE−BF)/(AE−DB)→X
:(AF−CD)/(AE−DB)→Y
:Disp X
:Disp Y
:End
:Lbl 1
:Disp "NO UNIQUE SOLUTION"
:End
```

Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a 2×3 matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the y-intercepts of the lines are between -10 and 10.)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. To use this program, dimension matrix [A] as a 2×3 matrix. Press ENTER after each screen display to continue the program.

```
Prgm4:ROWOPS
:Disp "ENTER A"
:Disp "2 BY 3 MATRIX"
:Disp "A B C"
:Disp "D E F"
:Input A
:Input B
:Input C
:Input D
:Input E
:Input F
:A→[A](1,1)
:B→[A](1,2)
:C→[A](1,3)
:D→[A](2,1)
:E→[A](2,2)
:F→[A](2,3)
:ClrHome
:Disp "ORIGINAL MATRIX"
:Disp [A]
:Pause
:"B^-1(C-AX)"→Y2
:"E^-1(F-DX)"→Y1
:-10→Xmin
:10→Xmax
:1→Xscl
:-10→Ymin
:10→Ymax
:1→Yscl
:DispGraph
:Pause
:ClrHome
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 1"
:*row(A^-1,[A],1)→[A]
:Disp [A]
:Pause
:ClrDraw
:"(A/B)(C/A-X)"→Y2
:DispGraph
:Pause
:ClrHome
:Disp "OBTAIN 0 BELOW"
:Disp "LEADING 1 IN"
:Disp "COLUMN 1"
:*row+(-D,[A],1,2)→[A]
:Disp[A]
:Pause
:ClrDraw
:"(E-(BD/A))^-1(F-(DC/A))"→Y1
:DispGraph
:Pause
:ClrHome
:[A](2,2)→G
:If G=0
:Goto 1
:*row(G^-1,[A],2)→[A]
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 2"
:Disp [A]
:Pause
:ClrDraw
:DispGraph
:Pause
:ClrHome
:Disp "OBTAIN 0 ABOVE"
:Disp "LEADING 1 IN"
:Disp "COLUMN 2"
:[A](1,2)→H
:*row+(-H,[A],2,1)→[A]
:Disp [A]
:Pause
:ClrDraw
:Y2-Off
:Line([A](1,3),-10,[A](1,3),10)
:DispGraph
:Pause
:ClrHome
:Disp "THE POINT OF"
:Disp "INTERSECTION IS"
:Disp "X="
:Disp [A](1,3)
:Disp "Y="
:Disp [A](2,3)
:End
:Lbl 1
:If [A](2,3)=0
:Disp "INFINITELY MANY"
:Disp "SOLUTIONS"
:If [A](2,3)≠0
:Disp "INCONSISTENT"
:Disp "SYSTEM"
:End
```

Evaluating an Algebraic Expression Program

This program can be used to evaluate an algebraic expression in one variable at several values of the variable. To use this program, enter an expression in Y_1.

```
Prgm5:EVALUATE
:Lbl 1
:Disp "ENTER X"
:Input X
:Disp Y1
:Goto 1
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
Prgm6:ADDVECT
:ClrDraw
:Disp "ENTER(A,B)"
:Disp "ENTER A"
:Input A
:Disp "ENTER B"
:Input B
:Disp "ENTER (C,D)"
:Disp "ENTER C"
:Input C
:Disp "ENTER D"
:Input D
:Line(0,0,A,B)
:Line(0,0,C,D)
:A+C→E
:B+D→F
:Line(0,0,E,F)
:Line(A,B,E,F)
:Line(C,D,E,F)
:Pause
:End
```

Graphing a Sine Function Program

This program will simultaneously draw a unit circle and the corresponding points on the sine curve. After the circle and sine curve are drawn, you can connect the points on the unit circle with their corresponding points on the sine curve by pressing ENTER.

```
Prgm7:SINESHOW
:Rad
:ClrDraw
:Param
:Simul
:-2.25→Xmin
:π/2→Xmax
:3→Xscl
:-1.19→Ymin
:1.19→Ymax
:1→Yscl
:0→Tmin
:6.3→Tmax
:.15→Tstep
:"-1.25+cos T"→X1T
:"sin T"→Y1T
:"T/4"→X2T
:"sin T"→Y2T
:DispGraph
:1→N
:Lbl 1
:IS>(N,12)
:Goto 2
:Pause
:Function
:Sequence
:Disp " "
:End
:Lbl 2
:Nπ/6.5→T
:-1.25+cos T→A
:sin T→B
:T/4→C
:Line(A,B,C,B)
:Pause
:Goto 1
```

Finding the Angle Between Two Vectors Program

This program will sketch two vectors and calculate the measure of the angle between the vectors. Be sure to set an appropriate viewing rectangle.

```
Prgm8:VECANGL
:ClrHome
:Deg
:Disp "ENTER (A,B)"
:Disp "ENTER A"
:Input A
:Disp "ENTER B"
:Input B
:ClrHome
:Disp "ENTER (C,D)"
:Disp "ENTER C"
:Input C
:Disp "ENTER D"
:Input D
:Line(0,0,A,B)
:Line(0,0,C,D)
:Pause
:AC+BD→E
:√(A²+B²)→U
:√(C²+D²)→V
:cos⁻¹(E/(UV))→θ
:Disp "θ="
:Disp θ
:ClrDraw
:End
```

TI-82
TI-83

Quadratic Formula Program

This program will display the solutions of a quadratic equation or the words "No Real Solution." To use the program, write the quadratic equation in general form and enter the values of *a, b,* and *c*.

```
PROGRAM:QUADRAT
:Disp "AX²+BX+C=0"
:Prompt A
:Prompt B
:Prompt C
:B²−4AC→D
:If D≥0
:Then
:(-B+√(D))/(2A)→M
:Disp M
:(-B−√(D))/(2A)→N
:Disp N
:Else
:Disp "NO REAL SOLUTION"
:End
```

Graph Reflection Program

This program will graph a function *f* and its reflection in the line $y = x$. To use this program, enter the function in Y_1 and set a viewing rectangle.

```
PROGRAM:REFLECT
:63Xmin/95→Ymin
:63Xmax/95→Ymax
:Xscl→Yscl
:"X"→Y₂
:DispGraph
:(Xmax−Xmin)/94→I
:Xmin→X
:While X≤Xmax
:Pt-On(Y₁,X)
:X+I→X
:End
```

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
PROGRAM:SOLVE
:Disp "AX+BY=C"
:Prompt A
:Prompt B
:Prompt C
:Disp "DX+EY=F"
:Prompt D
:Prompt E
:Prompt F
:If AE−DB=0
:Then
:Disp "NO UNIQUE"
:Disp "SOLUTION"
:Else
:(CE−BF)/(AE−DB)→X
:(AF−CD)/(AE−DB)→Y
:Disp X
:Disp Y
:End
```

Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a 2×3 matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the y-intercepts of the lines are between -10 and 10.)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. To use this program, dimension matrix [A] as a 2×3 matrix. Press ENTER after each screen display to continue the program.

```
PROGRAM: ROWOPS
:Disp "ENTER A"
:Disp "2 BY 3 MATRIX:"
:Disp "A B C"
:Disp "D E F"
:Prompt A,B,C
:Prompt D,E,F
:A→[A](1,1):B→[A](1,2)
:C→[A](1,3):D→[A](2,1)
:E→[A](2,2):F→[A](2,3)
:ClrHome
:Disp "ORIGINAL MATRIX:"
:Pause [A]
:"B^-1(C-AX)"→Y2
:"E^-1(F-DX)"→Y1
:ZStandard:Pause:ClrHome
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 1"
:*row(A^-1,[A],1)→[A]
:Pause [A]:ClrDraw
:"(A/B)(C/A-X)"→Y2
:DispGraph:Pause:ClrHome
:Disp "OBTAIN 0 BELOW"
:Disp "LEADING 1 IN"
:Disp "COLUMN 1"
:*row+(-D,[A],1,2)→[A]
:Pause [A]:ClrDraw
:"(E-(BD/A))^-1(F-(DC/A))"→Y1
:DispGraph:Pause:ClrHome
:[A](2,2)→G
:If G=0
:Goto 1
:*row(G^-1,[A],2)→[A]
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 2"
:Pause [A]:ClrDraw
:DispGraph:Pause:ClrHome
:Disp "OBTAIN 0 ABOVE"
:Disp "LEADING 1 IN"
:Disp "COLUMN 2"
:[A](1,2)→H
:*row+(-H,[A],2,1)→[A]
:Pause [A]:ClrDraw:FnOff 2
:Vertical -(B/A)(E-(BD/A))^-1(F-DC/A)+C/A
:DispGraph:Pause:ClrHome
:Disp "THE POINT OF"
:Disp "INTERSECTION IS"
:Disp "X=",[A](1,3),"Y=",[A](2,3)
:Stop
:Lbl 1
:If [A](2,3)=0
:Then
:Disp "INFINITELY MANY"
:Disp "SOLUTIONS"
:Else
:Disp "INCONSISTENT"
:Disp "SYSTEM"
:End
```

Evaluating an Algebraic Expression Program

This program can be used to evaluate an algebraic expression in one variable at several values of the variable. To use this program, enter an expression in Y_1.

```
PROGRAM:EVALUATE
:Lbl A
:Input "ENTER X",X
:Disp Y1
:Goto A
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
PROGRAM:ADDVECT
:ClrDraw
:Input "ENTER A",A
:Input "ENTER B",B
:Input "ENTER C",C
:Input "ENTER D",D
:Line(0,0,A,B)
:Line(0,0,C,D)
:A+C→E
:B+D→F
:Line(0,0,E,F)
:Line(A,B,E,F)
:Line(C,D,E,F)
:Pause
:Stop
```

Graphing a Sine Function Program

This program will simultaneously draw a unit circle and the corresponding points on the sine curve. After the circle and sine curve are drawn, you can connect the points on the unit circle with their corresponding points on the sine curve by pressing ENTER.

```
PROGRAM:SINESHOW
:Radian
:ClrDraw:FnOff
:Param:Simul
:-2.25→Xmin
:π/2→Xmax
:3→Xscl
:-1.19→Ymin
:1.19→Ymax
:1→Yscl
:0→Tmin
:6.3→Tmax
:.15→Tstep
:"-1.25+cos (T)"→X1T
:"sin (T)"→Y1T
:"T/4"→X2T
:"sin (T)"→Y2T
:DispGraph
:For(N,1,12)
:Nπ/6.5→T
:-1.25+cos (T)→A
:sin(T)→B
:T/4→C
:Line(A,B,C,B)
:Pause
:End
:Pause :Func
:Sequential:Disp
```

Finding the Angle Between Two Vectors Program

This program will sketch two vectors and calculate the measure of the angle between the vectors. Be sure to set an appropriate viewing rectangle.

```
PROGRAM:VECANGL
:ClrHome
:Degree
:Disp "ENTER (A,B)"
:Input "ENTER A",A
:Input "ENTER B",B
:ClrHome
:Disp "ENTER (C,D)"
:Input "ENTER C",C
:Input "ENTER D",D
:Line(0,0,A,B)
:Line(0,0,C,D)
:Pause
:AC+BD→E
:√(A²+B²)→U
:√(C²+D²)→V
:cos⁻¹(E/(UV))→θ
:ClrDraw:ClrHome
:Disp "θ=",θ
:Stop
```

TI-85
TI-86

Quadratic Formula Program

This program will display the solutions of a quadratic equation or the words "No Real Solution." To use the program, write the quadratic equation in general form and enter the values of *a*, *b*, and *c*. This program gives both real and complex answers. Solutions of a quadratic equation are also available directly by using the POLY function.

```
PROGRAM:QUADRAT
:Disp "AX²+BX+C=0"
:Input "ENTER A",A
:Input "ENTER B",B
:Input "ENTER C",C
:B²−4*A*C→D
:(-B+√(D))/(2A)→M
:Disp M
:(-B−√(D))/(2A)→N
:Disp N
```

Graph Reflection Program

This program will graph a function *f* and its reflection in the line $y = x$. To use this program, enter the function in y1 and set a viewing rectangle.

```
PROGRAM:REFLECT
:63*xMin/127→yMin
:63*xMax/127→yMax
:xScl→yScl
:y2=x
:DispG
:(xMax−xMin)/126→I
:xMin→x
:Lbl A
:PtOn(y1,x)
:x+I→x
:If x>xMax
:Stop
:Goto A
```

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
PROGRAM:SOLVE
:Disp "AX+BY=C"
:Input "ENTER A",A
:Input "ENTER B",B
:Input "ENTER C",C
:Disp "DX+EY=F"
:Input "ENTER D",D
:Input "ENTER E",E
:Input "ENTER F",F
:If A*E−D*B==0
:Goto A
:(C*E−B*F)/(A*E−D*B)→X
:(A*F−C*D)/(A*E−D*B)→Y
:Disp X
:Disp Y
:Stop
:Lbl A
:Disp "NO UNIQUE SOLUTION"
```

Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a 2×3 matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the y-intercepts of the lines are between -10 and 10.)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. To use this program, dimension matrix TEMP as a 2×3 matrix. Press ENTER after each screen display to continue the program.

```
PROGRAM: ROWOPS
:Disp "ENTER A"
:Disp "2 BY 3 MATRIX:"
:Disp "A B C"
:Disp "D E F"
:Prompt A,B,C
:Prompt D,E,F
:A→TEMP(1,1):B→TEMP(1,2)
:C→TEMP(1,3):D→TEMP(2,1)
:E→TEMP(2,2):F→TEMP(2,3)
:ClLCD
:Disp "ORIGINAL MATRIX:"
:Disp TEMP
:Pause
:y2=B^-1(C-A*x)
:y1=E^-1(F-D*x)
:ZStd:Pause:ClLCD
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 1"
:multR(A^-1,TEMP,1)→TEMP
:Disp TEMP:Pause
:"(A/B)(C/A-x)"→y
:ClDrw:DispG:Pause:ClLCD
:Disp "OBTAIN 0 BELOW"
:Disp "LEADING 1 IN"
:Disp "COLUMN 1"
:mRAdd(-D,TEMP,1,2)→TEMP
:Disp TEMP:Pause:
:If TEMP(2,2)==0
:Goto A
:y1=(E-(B*D/A))^-1(F-(D*C/A))
:ClDrw:DispG:Pause:ClLCD
:TEMP(2,2)→G
:multR(G^-1,TEMP,2)→TEMP
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 2"
:Disp TEMP:Pause:ClDrw
:DispG:Pause:ClLCD
:Disp "OBTAIN 0 ABOVE"
:Disp "LEADING 1 IN"
:Disp "COLUMN 2"
:TEMP(1,2)→H
:mRAdd(-H,TEMP,2,1)→TEMP
:Disp TEMP:Pause:FnOff 2:ClDrw
:Vert -(B/A)(E-(B*D/A))^-1(F-D*C/A)+C/A
:DispG:Pause:ClLCD
:Disp "THE POINT OF"
:Disp "INTERSECTION IS"
:Disp "X=",TEMP(1,3),"Y=",TEMP(2,3)
:Stop
:Lbl A
:If TEMP(2,3)==0
:Then
:Disp "INFINITELY MANY"
:Disp "SOLUTIONS"
:Else
:Disp "INCONSISTENT"
:Disp "SYSTEM"
:End
```

Evaluating an Algebraic Expression Program

This program can be used to evaluate an algebraic expression in one variable at several values of the variable. To use this program, enter an expression in y1.

```
PROGRAM:EVALUATE
:Lbl A
:Input "Enter x",x
:Disp y1
:Goto A
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
PROGRAM:ADDVECT
:ClDrw
:Input "enter A",A
:Input "enter B",B
:Input "enter C",C
:Input "enter D",D
:Line(0,0,A,B)
:Line(0,0,C,D)
:A+C→E
:B+D→F
:Line(0,0,E,F)
:Line(A,B,E,F)
:Line(C,D,E,F)
:Pause
:Stop
```

Graphing a Sine Function Program

This program will simultaneously draw a unit circle and the corresponding points on the sine curve. After the circle and sine curve are drawn, you can connect the points on the unit circle with their corresponding points on the sine curve by pressing ENTER.

```
PROGRAM:SINESHOW
:Radian
:ClDrw:FnOff
:Param:SimulG
:-2.25→xMin
:π/2→xMax
:3→xScl
:-1.1→yMin
:1.1→yMax
:1→yScl
:0→tMin
:6.3→tMax
:.15→tStep
:xt1=-1.25+cos t
:yt1=sin t
:xt2=t/4
:yt2=sin t
:For(N,1,12)
:N*π/6.5→t
:-1.25+cos t→A
:sin t→B
:t/4→C
:Line(A,B,C,B)
:Pause
:End
:Pause :Func
:SeqG:Disp
```

Finding the Angle Between Two Vectors Program

This program will sketch two vectors and calculate the measure of the angle between the vectors. Be sure to set an appropriate viewing rectangle.

```
PROGRAM:VECANGL
:ClLCD
:Degree
:Disp "Enter (A,B)"
:Input "Enter A",A
:Input "Enter B",B
:ClLCD
:Disp "Enter (C,D)"
:Input "Enter C",C
:Input "Enter D",D
:Line(0,0,A,B)
:Line(0,0,C,D)
:Pause
:A*C+B*D→E
:√(A²+B²)→U
:√(C²+D²)→V
:cos⁻¹(E/(U*V))→θ
:ClDrw:ClLCD
:Disp "θ=",θ
:Stop
```

TI-92

Quadratic Formula Program

This program will display the solutions of a quadratic equation or the words "No Real Solution." To use the program, write the quadratic equation in general form and enter the values of *a, b,* and *c*. This program gives both real and complex answers.

```
:quadrat( )
:Prgm
:setMode("Complex Format","RECTANGULAR")
:Disp "AX^2+BX+C=0"
:Input "Enter A.",a
:Input "Enter B.",b
:Input "Enter C.",c
:b^2-4*a*c→d
:(-b+√(d))/(2*a)→m
:(-b-√(d))/(2*a)→n
:Disp m
:Disp n
:setMode("Complex Format","REAL")
:EndPrgm
```

Graph Reflection Program

This program will graph a function *f* and its reflection in the line $y = x$. To use this program, enter the function in y1 and set a viewing rectangle.

```
:reflect ( )
:Prgm
:103xmin/239→ymin
:103xmax/239→ymax
:xscl→yscl
:x→y2(x)
:DispG
:(xmax-xmin)/238→n
:xmin→x
:While x<xmax
:   PtOn y1(x),x
:   x+n→x
:EndWhile
:EndPrgm
```

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
:solvelin( )
:Prgm
:ClrIO
:Disp "Ax+By=C"
:Input "Enter A.",a
:Input "Enter B.",b
:Input "Enter C.",c
:ClrIO
:Disp "Dx+Ey=F"
:Input "Enter D.",d
:Input "Enter E.",e
:Input "Enter F.",f
:If a*e-d*b=0 Then
:      Disp "No unique solution"
:   Else
:      (c*e-b*f)/(a*e-d*b)→x
:      (a*f-c*d)/(a*e-d*b)→y
:      Disp x
:      Disp y
:EndIf
:EndPrgm
```

Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a 2×3 matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the y-intercepts of the lines are between -10 and 10.)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. Press ENTER after each screen display to continue the program.

```
:rowops( )
:Prgm
:ClrIO
:ClrHome
:setMode("Split Screen","Left-Right")
:setMode("Split 1 App","Home")
:setMode("Split 2 App","Graph")
:Disp "ENTER A"
:Disp "2 BY 3 MATRIX:"
:Disp "A B C"
:Disp "D E F"
:Prompt a,b,c
:Prompt d,e,f
:[[a,b,c][d,e,f]]→mat1
:ClrIO
:b^(-1)*(c−a*x)→y2(x)
:e^(-1)*(f−d*x)→y1(x)
:ZoomStd
:Disp "ORIGINAL MATRIX:"
:Pause mat1
:ClrIO
:a/b*(c/a−x)→y2(x)
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 1"
:mRow(a^(-1),mat1,1)→mat1
:Pause mat1
:ClrIO
:(e−b*d/a)^(-1)*(f−d*c/a)→y1(x)
:DispG
:Disp "OBTAIN 0 BELOW"
:Disp "LEADING 1 IN"
:Disp "COLUMN 1"
:mRowAdd(-d,mat1,1,2)→mat1
:Pause mat1
:ClrIO
:mat1[2,2]→g
:If g=0
:Goto a1
:Disp "OBTAIN LEADING"
:Disp "1 IN ROW 2"
:mRow(g^(-1),mat1,2)→mat1
:Pause mat1
:ClrIO
:mat1[1,2]→h
:FnOff 2
:LineVert -b/a*(e−b*d/a)^(-1)*(f−d*c/a)+c/a
:Disp "OBTAIN 0 ABOVE"
:Disp "LEADING 1 IN"
:Disp "COLUMN 2"
:mRowAdd(-h,mat1,2,1)→mat1
:Pause mat1
:ClrIO
:Disp "THE POINT OF"
:Disp "INTERSECTION IS"
:Disp "X=",mat1[1,3],"Y=",mat1[2,3]
:Goto A2
:Lbl a1
:If mat1[2,3]=0 Then
:Disp "INFINITELY MANY"
:Disp "SOLUTIONS"
:Else
:Disp "INCONSISTENT"
:Disp "SYSTEM"
:EndIf
:Lbl A2
:Pause
:setMode("Split Screen","Full")
:EndPrgm
```

Evaluating an Algebraic Expression Program

This program can be used to evaluate an algebraic expression in one variable at several values of the variable. To use this program, enter an expression in y1.

```
:evaluate( )
:Prgm
:Lbl one
:Input "enter x",x
:Disp y1(x)
:Goto one
:EndPrgm
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
:addvect( )
:Prgm
:ClrIO
:Input "ENTER a",a
:Input "ENTER b",b
:Input "ENTER c",c
:Input "ENTER d",d
:ClrDraw
:Line(0,0,a,b)
:Line(0,0,c,d)
:a+c→e
:b+d→f
:Line 0,0,e,f
:Line a,b,e,f
:Line c,d,e,f
:Pause
:setMode("Split 1 App","Home")
:Stop
:EndPrgm
```

Graphing a Sine Function Program

This program will simultaneously draw a unit circle and the corresponding points on the sine curve. After the circle and sine curve are drawn, you can connect the points on the unit circle with their corresponding points on the sine curve by pressing ENTER.

```
:sineshow( )
:Prgm
:Disp
:ClrDraw:FnOff
:setMode("Graph","Parametric")
:setGraph("Graph Order","Simul")
:-2.9→xmin
:3π/4→xmax
:3→xscl
:-1.1→ymin
:1.1→ymax
:1→yscl
:0→tmin
:6.3→tmax
:.15→tstep
:-1.25+cos(t)→xt1(t)
:sin(t)→yt1(t)
:t/4→xt2(t)
:sin(t)→yt2(t)
:DispG
:For N,1,12
:N*π/6.5→t
:-1.25+cos(t)→A
:sin(t)→B
:t/4→C
:Line A,B,C,B
:Pause
:EndFor
:Pause
:setMode("Graph","Function")
:setGraph("Graph order","Seq")
:setMode("Split 1 App","Home")
:EndPrgm
```

Finding the Angle Between Two Vectors Program

This program will sketch two vectors and calculate the measure of the angle between the vectors. Be sure to set an appropriate viewing rectangle.

```
:vecangl( )
:Prgm
:FnOff
:ClrHome:ClrDraw
:SetMode("Split Screen","Left-Right")
:SetMode("Split 1 App","Home")
:SetMode("Split 2 App","Graph")
:SetMode("Exact/Approx","Approximate")
:ClrIO
:Disp "ENTER (A,B)"
:Input "ENTER A",A
:Input "ENTER B",B
:Line(0,0,A,B)
:Pause
:ClrIO
:Disp "ENTER (C,D)"
:Input "ENTER C",C
:Input "ENTER D",D
:Line(0,0,C,D)
:Pause
:ClrIO
:A*C+B*D→E
:√((A^2+B^2))→U
:√(C^2+D^2)→V
:cos^-1 (E/(U*V))→θ
:Disp "θ=",θ
:Pause
:SetMode("Exact/Approx","Auto")
:SetMode("Split Screen","Full")
:SetMode("Split 1 App","Home")
:Stop
:EndPrgm
```

Casio fx-7700G

Quadratic Formula Program

This program will display the solutions of a quadratic equation or the words "No Real Solution." To use the program, write the quadratic equation in general form and enter the values of *a, b,* and *c*.

```
QUADRAT
"AX²+BX+C=0"
"A="?→A
"B="?→B
"C="?→C
B²−4AC→D
D<0⇒Goto 1
"X=":(-B+√D)÷(2A)◢
"OR X=":(-B−√D)÷(2A)
Goto 2
Lbl 1
"NO REAL SOLUTION"
Lbl 2
```

Graph Reflection Program

This program will graph a function *f* and its reflection in the line $y = x$. To use this program, enter the function in f1.

```
REFLECT
"GRAPH -A TO A"
"A="?→A
Range -A,A,1,-2A÷3,2A÷3,1
Graph Y=f1
-A→B
Lbl 1
B→X
Plot f1,B
B+A÷32→B
B≤A⇒Goto1 :Graph Y=X
```

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

```
SOLVE
"AX+BY=C"
"A="?→A
"B="?→B
"C="?→C
"DX+EY=F"
"D="?→D
"E="?→E
"F="?→F
AE−DB=0⇒Goto 1
"X=":(CE−BF)÷(AE−DB)◢
"Y=":(AF−CD)÷(AE−DB)
Goto 2
Lbl 1
"NO UNIQUE SOLUTION"
Lbl 2
```

Visualizing Row Operations Program not available

Evaluating an Algebraic Expression Program

This program can be used to evaluate an algebraic expression in one variable at several values of the variable. To use this program, enter an expression in f1.

```
EVALUATE
Lbl 1
"X="?→X
"F(X)=" : f1◢
Goto 1
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
ADDVECT
Cls
"A="?→A
"B="?→B
"C="?→C
"D="?→D
Plot 0,0
Plot A,B
Line
Plot 0,0
Plot C,D
Line◢
A+C→E
B+D→F
Plot 0,0
Plot E,F
Line
Plot A,B
Plot E,F
Line
Plot C,D
Plot E,F
Line◢
```

Graphing a Sine Function Program

This program will simultaneously draw a unit circle and the corresponding points on the sine curve. After the circle and sine curve are drawn, you can connect the points on the unit circle with their corresponding points on the sine curve by pressing [EXE]. Press Mode Shift X to change to parametric mode when starting to write this program.

```
SINESHOW
Rad
Range -2.25,π÷2,3,-1.19,1.19,10,6.3,.15
Graph(X,Y)=(-1.25+cos T,sinT)
Graph(X,Y)=(T÷4,sinT)
0→N
Lbl 1
N+1→N
Nπ÷6.5→T
-1.25+cos T→A
sin T→B
T÷4→C
Plot A,B
Plot C,B
Line◢
N<12⇒Goto 1
```

Finding the Angle Between Two Vectors Program

This program will sketch two vectors and calculate the measure of the angle between the vectors. Be sure to set an appropriate viewing rectangle.

```
VECANGL
Cls
Deg
"ENTER (A,B)"
"A="?→A
"B="?→B
"ENTER (C,D)"
"C="?→C
"D="?→D
Plot 0,0
Plot A,B
Line
Plot 0,0
Plot C,D
Line◢
AC+BD→E
√(A²+B²)→U
√(C²+D²)→V
cos⁻¹(E÷UV)→θ
"θ="
θ
```

Casio fx-7700GE
Casio fx-9700GE
Casio CFX-9800G
Casio CFX-9850G

Quadratic Formula Program

This program will display the solutions of a quadratic equation or the words "No Real Solution." To use the program, write the quadratic equation in general form and enter the values of *a, b,* and *c*.

Casio fx-7700GE

Solutions to quadratic equations are also available directly from the Casio calculator's EQUATION MODE.

```
QUADRAT
"AX²+BX+C=0"↵
"A="?→A↵
"B="?→B↵
"C="?→C↵
B²−4AC→D↵
D<0⇒Goto 1↵
(-B+√(D)÷(2A)◢
(-B−√(D)÷(2A)↵
Goto 2↵
Lbl 1↵
"NO REAL SOLUTION"↵
Lbl 2
```

Casio fx-9700GE
Casio CFX-9800G
Casio CFX-9850G

Both real and complex answers are given. Solutions to quadratic equations are also available directly from the Casio calculator's EQUATION MODE.

```
QUADRAT
"AX²+BX+C=0"↵
"A="?→A↵
"B="?→B↵
"C="?→C↵
B²−4AC→D↵
(-B+√(D)÷(2A)◢
(-B−√(D)÷(2A)
```

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists. Solutions to systems of linear equations are also available directly from the Casio calculator's EQUATION MENU.

```
SOLVE
"AX+BY=C"↵
"A=":?→A↵
"B=":?→B↵
"C=":?→C↵
"DX+EY=F"↵
"D=":?→D↵
"E=":?→E↵
"F=":?→F↵
AE−DB=0⇒Goto 1↵
"X=":(CE−BF)÷(AE−DB)◢
"Y=":(AF−CD)÷(AE−DB)↵
Goto 2↵
Lbl 1↵
"NO UNIQUE SOLUTION"↵
Lbl 2
```

Graph Reflection Program

This program will graph a function *f* and its reflection in the line $y = x$. To use this program, enter the function in f1.

Casio fx-7700GE

To use this program, enter the function in f1.

```
REFLECT
"GRAPH -A TO A"↵
"A="?→A↵
Range -A,A,1,-2A÷3,2A÷3,1↵
Graph Y=f1↵
-A→B↵
Lbl 1↵
B→X↵
Plot f1,B↵
B+A÷32→B↵
B≤A⇒Goto1:Graph Y=X
```

Casio fx-9700GE

To use this program, enter a function in f1 and set a viewing rectangle.

```
REFLECT
63Xmin÷127→A↵
63Xmax÷127→B↵
Xscl→C↵
Range , , , A, B, C↵
(Xmax−Xmin)÷126→I↵
Xmax→M↵
Xmin→D↵
Graph Y=f1↵
Lbl 1↵
D→X↵
Plot f1,D↵
D+I→D↵
D≤M⇒Goto 1:Graph Y=X
```

Casio CFX-9800G

To use this program, enter a function in f1 and set a viewing rectangle.

```
REFLECT
63Xmin÷95→A↵
63Xmax÷95→B↵
Xscl→C↵
Range , , , A, B, C↵
(Xmax−Xmin)÷94→I↵
Xmax→M↵
Xmin→D↵
Graph Y=f1↵
Lbl 1↵
D→X↵
Plot f1,D↵
D+I→D↵
D≤M⇒Goto 1:Graph Y=X
```

Casio CFX-9850G

Use the program for the Casio fx-9700GE and replace the line "Range , , , A,B,C,↵" with
"View Window , , , A,B,C,↵."

Visualizing Row Operations Program

This program demonstrates how elementary matrix row operations used in Gauss-Jordan elimination may be interpreted graphically. It asks the user to enter a 2×3 matrix that corresponds to a system of two linear equations. (The matrix entries should not be equivalent to either vertical or horizontal lines. This demonstration is also most effective if the y-intercepts of the lines are between -10 and 10.)

While the demonstration is running, you should notice that each elementary row operation creates an equivalent system. This equivalence is reinforced graphically because, although the equations of the lines change with each elementary row operation, the point of intersection remains the same. You may want to run this program a second time to notice the relationship between the row operations and the graphs of the lines of the system. To use this program, dimension Mat A as a 2×3 matrix. Press [EXE] after each screen display to continue the program.

```
ROWOPS
"ENTER A"↵
"2 BY 3 MATRIX:"↵
"A B C"↵
"D E F"↵
"A="?→A:"B="?→B:
"C="?→C:"D="?→D:
"E="?→E:"F="?→F:↵
[[A,B,C][D,E,F]]→Mat A↵
Cls↵
"ORIGINAL MATRIX:"◢
Mat A◢
Range -10,10,1,-10,10,1↵
Graph Y=B^-1(C−AX)↵
Graph Y=E^-1(F−DX)◢
Cls↵
"OBTAIN LEADING"↵
"1 IN ROW 1"◢
∗Row A^-1,A,1↵
Mat A◢
Graph Y=(A÷B)(C÷A−X)↵
Graph Y=E^-1(F−DX)◢
Cls↵
"OBTAIN 0 BELOW"↵
"LEADING 1 IN"↵
"COLUMN 1"◢
∗Row+ -D,A,1,2↵
Mat A◢
Graph Y=(A÷B)(C÷A−X)↵
GraphY=(E−(BD÷A))^-1(F−(DC÷A))◢
Cls↵
Mat A[2,2]→G↵
G=0⇒Goto 1↵
∗Row G^-1,A,2↵
"OBTAIN LEADING"↵
"1 IN ROW 2"◢
Mat A◢
Graph Y=(A÷B)(C÷A−X)↵
Graph Y=(E−(BD÷A))^-1(F−(DC÷A))◢
Cls↵
"OBTAIN 0 ABOVE"↵
"LEADING 1 IN"↵
"COLUMN 2"◢
Mat A[1,2]→H↵
∗Row+ -H,A,2,1↵
Mat A◢
Mat A[1,3]→J↵
Mat A[2,3]→K↵
Graph Y=K↵
Plot J,-10:Plot J,10:Line◢
"THE POINT OF"↵
"INTERSECTION IS"↵
"X=":J◢
"Y=":K
Goto 3↵
Lbl 1↵
Mat A[2,3]=0⇒Goto 2↵
"INCONSISTENT"↵
"SYSTEM"↵
Goto 3↵
Lbl 2↵
"INFINITELY MANY"↵
"SOLUTIONS"↵
Lbl 3
```

Evaluating an Algebraic Expression Program

This program can be used to evaluate an algebraic expression in one variable at several values of the variable. To use this program, enter an expression in f1.

```
EVALUATE
Lbl 1↵
"X="?→X↵
"F(X)=" : f1◢
Goto 1
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
ADDVECT
Cls↵
"A="?→A↵
"B="?→B↵
"C="?→C↵
"D="?→D↵
Plot 0,0↵
Plot A,B↵
Line↵
Plot 0,0↵
Plot C,D↵
Line↵
A+C→E↵
B+D→F↵
Plot 0,0↵
Plot E,F↵
Line↵
Plot A,B↵
Plot E,F↵
Line↵
Plot C,D↵
Plot E,F↵
Line◢
```

Graphing a Sine Function Program

This program will simultaneously draw a unit circle and the corresponding points on the sine curve. After the circle and sine curve are drawn, you can connect the points on the unit circle with their corresponding points on the sine curve by pressing [EXE]. When starting to write this program, press SHIFT SET UP and select PRM or PARM for the GRAPH TYPE to change to parametric mode.

```
SINESHOW
Rad↵
Range -2.25,π÷2,3,-1.19,1.19,1,0,6.3,.15↵
Graph(X,Y)=(-1.25+cos T,sin T)↵
Graph(X,Y)=(T÷4,sin T)↵
0→N↵
Lbl 1↵
N+1→N↵
Nπ÷6.5→T↵
-1.25+cosT→A↵
sin T→B↵
T÷4→C↵
Plot A,B↵
Plot C,B↵
Line◢
N<12⇒Goto 1↵
Cls
```

Casio CFX-9850G

Use the program for the Casio CFX-9800G and replace the line
"Range -2.25,π÷2,3,-1.19,1.19,1,0,6.3,.15↵"
with the line
"View Window -2.25,π÷2,3,-1.19,1.19,1,0,6.3,.15↵."

Finding the Angle Between Two Vectors Program

This program will sketch two vectors and calculate the measure of the angle between the vectors. Be sure to set an appropriate viewing rectangle.

```
VECANGL
Cls↵
Deg↵
"ENTER (A,B)"↵
"A="?→A↵
"B="?→B↵
"ENTER (C,D)"↵
"C="?→C↵
"D="?→D↵
Plot 0,0↵
Plot A,B↵
Line↵
Plot 0,0↵
Plot C,D↵
Line◢
AC+BD→E↵
√(A²+B²)→U↵
√(C²+D²)→V↵
cos⁻¹(E÷UV)→θ↵
"θ="↵
θ
```

Sharp EL-9200C
Sharp EL-9300C

Quadratic Formula Program

This program will display the solutions of a quadratic equation or the words "No Real Solution." To use the program, write the quadratic equation in general form and enter the values of *a, b,* and *c*. This program gives both real and complex answers.

```
quadratic
————————COMPLEX
Print "ax²+bx+c=0"
Input a
Input b
Input c
d=b²−4a∗c
x1=(-b+√d)/(2a)
x2=(-b−√d)/(2a)
Print x1
Print x2
End
```

Graph Reflection Program

This program will graph a function *f* and its reflection in the line $y = x$. To use this program, replace f(X) with your expression in X.

```
reflection
————————REAL
Goto top
Label equation
Y=f(X)
Return
Label rng
xmin=-10
xmax=10
xstp=(xmax2xmin)/10
ymin=2xmin/3
ymax=2xmax/3
ystp=xstp
Range xmin,xmax,xstp,ymin,ymax,ystp
Return
Label top
Gosub rng
Graph X
step=(xmax−xmin)/(94∗2)
X=xmin
Label 1
Gosub equation
Plot X,Y
Plot Y,X
X=X+step
If X<=xmax Goto 1
End
```

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists. Equations must be entered in the form: AX + BY = C; DX + EY = F. Uppercase letters are used so that the values can be accessed in the calculation mode of the calculator.

```
solve
————————REAL
Print "AX+BY=C"
Input A
Input B
Input C
Print "DX+EY=F"
Input D
Input E
Input F
If A∗E−D∗B=0 Goto 1
X=(C∗E−B∗F)/(A∗E−D∗B)
Y=(A∗F−C∗D)/(A∗E−D∗B)
Print X
Print Y
End
Label 1
Print "no unique solution"
End
```

Visualizing Row Operations Program not available

Evaluating an Algebraic Expression Program

This program can be used to evaluate an algebraic expression in one variable at several values of the variable. To use this program, replace f(X) with your expression in X.

```
evaluate
————REAL
Goto top
Label equation
Y=f(X)
Return
Label top
Input X
Gosub equation
Print Y
Goto top
End
```

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle.

```
addvect
————REAL
ClrG
Input a
Input b
Input c
Input d
Line 0,0,a,b
Line 0,0,c,d
e=a+c
f=b+d
Line 0,0,e,f
Line a,b,e,f
Line c,d,e,f
Wait
End
```

Graphing a Sine Function Program

This program will simultaneously draw a unit circle and the corresponding points on the sine curve. After the circle and sine curve are drawn, you can connect the points on the unit circle with their corresponding points on the sine curve by pressing [ENTER].

```
sineshow
————REAL
m=sin^-1 1/(π/2)
Range -2.25,π/2,3,-1.19,1.19,1
step=π/15
θ=0
xco=-.25
xso=0
yo=0
Label 1
θ=θ+step
xc=cos(mθ)−1.25
xs=θ/4
y=sin (mθ)
Line xco,yo,xc,y
Line xso,yo,xs,y
xco=xc
xso=xs
yo=y
If θ<(2π) Goto 1
step=π/6
θ=0
Label 2
θ=θ+step
xc=cos (mθ)−1.25
xs=θ/4
y=sin (mθ)
Line xc,y,xs,y
Wait
If θ<2π Goto 2
End
```

Finding the Angle Between Two Vectors Program

This program will sketch two vectors and calculate the measure of the angle between the vectors. Be sure to set an appropriate viewing rectangle. Set the calculator to degree mode before running the program.

```
vecangl
————REAL
ClrG
ClrT
Print“enter (a,b)”
Input a
Input b
ClrT
Print“enter (c,d)”
Input c
Input d
Line 0,0,a,b
Line 0,0,c,d
Wait
e=a∗c+b∗d
u=√(a²+b²)
v=√(c²+d²)
t=cos⁻¹(e/(u∗v))
Print t
End
```

HP-38G

Quadratic Formula Program

This program will display the solutions of a quadratic equation or the words "No Real Solution." To use the program, write the quadratic equation in general form and enter the values of *a, b,* and *c*. This program displays the answer in complex form (x, y), where x is the real part and y is the imaginary part.

```
QUADRAT PROGRAM
INPUT A;"AX²+BX+C=0";
"ENTER A";"";1:
INPUT B;"AX²+BX+C=0";
"ENTER B";"";1:
INPUT C;"AX²+BX+C=0";
"ENTER C";"";1:
B²−4AC▶D:
(-B+√D)/(2A)▶Z1:
(-B+√D)/(2A)▶Z2:
DISP 3;Z1:
DISP 5;Z2:
FREEZE
```

Graph Reflection Program not available

Systems of Linear Equations Program

This program will display the solution of a system of two linear equations in two variables of the form

$$ax + by = c$$

$$dx + ey = f$$

if a unique solution exists.

1. Input the 2 programs SOLVE and SOLVE.SOLN.
2. Run the SOLVE program.

```
SOLVE
SOLVE PROGRAM
INPUT A;"AX+BY=C";
  "ENTER A";" ";1:
INPUT B;"AX+BY=C";
  "ENTER B";" ";1:
INPUT C;"AX+BY=C";
  "ENTER C";" ";1:
INPUT D;"DX+EY=F";
  "ENTER D";" ";1:
INPUT E;"DX+EY=F";
  "ENTER E";" ";1:
INPUT F;"DX+EY=F";
  "ENTER F";" ";1:
ERASE:
IF AE−DB==0
THEN DISP 3; "NO UNIQUE  SOLUTION":
ELSE RUN "SOLVE.SOLN":
END:
FREEZE:
SOLVE.SOLN PROGRAM
(CE−BF)/(AE−DB)▶X:
(AF−CD)/(AE−DB)▶Y:
DISP 3;"X="X:
DISP 5;"Y="Y:
```

Visualizing Row Operations Program not available

Evaluating an Algebraic Expression Program

This program can be used to evaluate an algebraic expression in one variable at several values of the variable. Use the Solve aplet to evaluate an expression.

1. Press LIB. Highlight the Solve aplet. Press {{START}}.
2. Set your expression equal to *y*, enter the equation (*y* = *your expression*) in E1, and press {{OK}}. The equation should be checked.
3. Press NUM.
4. Highlight the *x*-variable field. Enter a value for *x* and press {{OK}}.
5. Highlight the *y*-variable field and press {{SOLVE}}. The value of the expression will appear in the *y*-variable field.
6. Repeat steps 4 and 5 to evaluate the expression for other values of *x*.

Adding Vectors Graphically Program

This program will sketch two vectors in standard position. Using the parallelogram law for the vector addition, the program also sketches the vector sum. Be sure to set an appropriate viewing rectangle. The Function aplet should have a plot range of $-10 \le x \le 10$ and $-10 \le y \le 10$.

```
ADDVECT PROGRAM
INPUT A;; "ENTER A";;1:
INPUT B;; "ENTER B";;1:
INPUT C;; "ENTER C";;1:
INPUT D;; "ENTER D";;1:
ERASE:
LINE -10;0;10;0:
LINE 0;-10;0;10:
LINE 0;0;A;B:
LINE 0;0;C;D:
FREEZE:
A+C▶ E
B+D▶ F
LINE 0;0;E;F:
LINE A;B;E;F:
LINE C;D;E;F:
FREEZE
```

Graphing a Sine Function Program

This program will simultaneously draw a unit circle and the corresponding points on the sine curve. After the circle and sine curve are drawn, you can connect the points on the unit circle with their corresponding points on the sine curve by pressing ENTER.

1. Enter the 3 programs SINESHOW, DRAW.SINE, and DRAW.LINE.
2. Set the plot range in the Function aplet to $-3 \le x \le \pi/2$ and $-1.1 \le y \le 1.1$. Set the angle measure to radians.
3. Run the SINESHOW program.

```
SINESHOW PROGRAM
ASIN(1)/(π/2)▶M:
0▶T:
-.25▶A:
0▶B:
0▶C:
LINE -3;0;π/2;0:
LINE 0;-1.1;0;1.1:
FOR T=0 TO 31π/15
      STEP π/15;
      RUN "DRAW.SINE":
END:
0▶T:
FOR T=0 TO 2π
      STEP π/6;
      RUN "DRAW.LINE":
END
```

```
DRAW.SINE PROGRAM
COS(MT)−1.25▶D:
T/4▶E:
SIN(MT)▶F:
LINE A;C;D;F:
LINE B;C;E;F:
D▶A:
E▶B:
F▶C:
```

```
DRAW.LINE PROGRAM
COS(MT)−1.25▶D:
T/4▶E:
SIN(MT)▶F:
LINE D;F;E;F:
FREEZE
```

Finding the Angle Between Two Vectors Program

This program will sketch two vectors and calculate the measure of the angle between the vectors. Be sure to set an appropriate viewing rectangle. The Function aplet should have a plot range of $-10 \le x \le 10$ and $-10 \le y \le 10$. Set the MODE to degrees before running the program.

```
VECANGL PROGRAM
INPUT A; “ENTER (A,B)”; “ENTER A”;;1:
INPUT B; “ENTER (A,B)”; “ENTER B”;;1:
INPUT C; “ENTER (C,D)”; “ENTER C”;;1:
INPUT D; “ENTER (C,D)”; “ENTER D”;;1:
ERASE:
LINE-10;0;10;0:
LINE 0;-10;0;10:
LINE 0;0;A;B:
LINE 0;0;C;D:
FREEZE:
AC+BD▶E
√(A²+B²)▶U:
√(C²+D²)▶V:
ACOS(E/(UV))▶T:
ERASE:
DISP 3; “ANGLE = ”T:
FREEZE
```